
Abstract

The paper shows how natural numbers without any conceptual

limitation (i.e. arbitrarily sized) can be implemented and veri�ed in

Modern Ei�el. The fact that the implementation is highly ine�cient

to execute is accepted. The implementation shall serve as a model on

how to implement inductively de�ned data types in Modern Ei�el and

on how to proof important properties of the implemented functions.

Many correctness proof are given in full detail. Modern Ei�el's

proof engine can derive all these proofs mechanically. The developer

just has to state the properties.

1 Introduction

Inductively de�ned data types are an important subset of all data types.
The corresponding objects are immutable. Therefore reasoning about them
is simpler than reasoning about mutable objects. Induction can be used to
prove properties about functions of inductively de�ned types.

Modern Ei�el supports inductively de�ned data types. Its proof engine
can do many induction proofs automatically. Since the proof engine does
not try to do nested inductive proofs, the developer has to specify su�cient
intermediate assertions such that each asssertion requires just one inductive
proof to be veri�ed.

In this paper we describe the class MATH_NAT to model natural numbers.
The most important functions of naturals like successor, predescessor, plus,

minus, product, power and the functions to model equality and order relation
are implemented.

Many proofs are given in detail in order to demonstrate the capabilities
of Modern Ei�el's proof engine.

2 The class LINK

Our class MATH_NAT points to a linked chain of memory cells. The number
zero is represented by an empty chain, the number one is represented by a
chain with one element and so on. The memory cells do not have any content,
just a possibly void reference to the next cell. The linked cells are objects of
type LINK whose class text reads as

class LINK create
make_empty,
put_next

feature {NONE}
make_empty ensure nxt ˜ Void end
put_next(o: detachable like Current)

ensure nxt ˜ o end
feature

nxt: detachable like Current
end

Objects of type LINK have only one attribute which refers to the next link
in the chain. The attribute is detachable i.e. possibly void. The creation
procedures are su�ciently speci�ed because the speci�cation gives the value
of the attribute. We say the speci�cations of the creation procedures are
executable. Therefore no explicit implementation of the creation procedures
is necessary.

The operator ˜ is used as the identity operator in the following. If a˜b
is true then the objects attached to a and b are indistinguishable. If two
expressions e1 and e2 are indistiguishable i.e. if e1˜e2 is valid, the one can
be substituted by the other in any expression.

The veri�cation of the class LINK is trivial. The class has only two as-
sertions, one in each creation procedure. Since the implementation is derived
from the asssertions, the assertions are valid by de�nition.

3 Outline of the class MATH NAT

With the help of the class LINK we can de�ne the outline of our class
MATH_NAT.

immutable expanded class
MATH_NAT

create
make_zero

create {MATH_NAT}
from_link

feature {NONE}
rep: detachable immutable LINK -- the only attribute

feature
is_zero: BOOLEAN

ensure{NONE}
Result = (rep ˜ Void)

ensure
Result => Current ˜ MATH_NAT

end
...
invariant

... -- invariant properties
end

The class MATH_NAT is immutable and expanded1. The property of being
expanded means that objects of that type have no own identity. The identity
of these objects is completely de�ned by the identity of their attributes. The
class has only one attribute which is detachable, i.e. it can be void. This
makes it possible that there are initial objects which are indistinguishable.
In order to achieve that we de�ne the creation procedure make_zero as

feature {NONE}
make_zero

ensure{NONE}
rep ˜ Void

ensure
is_zero
Current ˜ MATH_NAT

end

1Expanded types have copy semantics and reference types have reference semantics.

Expanded type objects are never allocated on the heap. They reside either on the stack

or are subobjects of other objects

This procedure has three assertions. The �rst one is a private postcondition
which is an executable speci�cation of the routine (i.e. it speci�es the value
of the only attribute). Since the body is derived from that speci�cation, this
assertion is valid by de�nition.

The second postcondition states that the newly constructed object is_zero.
Expanding the de�nition of is_zero gives rep˜Void which is implied by the
�rst postcondition.

The third postcondition states that the newly constructed object is in-
distinguishable (i.e. identical) from any other newly constructed object of
the same type with the same creation procedure. This can be proved by the
fact that the class is expanded (i.e. it has no own identity) and by the fact
that its only attribute is of reference type and Void. Since two void values
are always identical, there is no possibility two distinguish two newly created
objects of type MATH_NAT.

Now we recall that the function is_zero has postconditions which still
need to be proved. The �rst postcondition is its executable de�nition and
therefore valid implicitely. The second postcondition states that any object
which is zero is indistinguishable from the initial object. Having a de�nition
of the initial object we can now prove this fact. The class MATH_NAT has only
one reference attribute. The default creation procedure make_zeromakes this
attribute void and the function is_zero returns true only if this attribute
is void. Therefore any natural a with a.is_zero is identical with the initial
object.

The second creation procedure is not publicly available. It can be used
internally to create other naturals.

from_link(lnk: immutable LINK)
ensure{NONE}

rep ˜ lnk
ensure

Current ˜ MATH_NAT(lnk)
is_zero = (lnk ˜ Void)

end

The postconditions are proved by the same arguments as above.
The creation procedure from_link is private to ensure that the links

represent really a linear chain. If this creation procedure were available
publicly, a client could create some links which are linked in a circular fashion
and create a natural with from_link which would not be consistent with the
naturals we want to create.

Up to now we can create only zeros. We come later to functions which
create non zero naturals. But before that we de�ne a predecessor function.

pred: MATH_NAT
require

not is_zero
ensure{NONE}

Result ˜ MATH_NAT(rep.nxt) -- executable specification
ensure

Result ˜ pred -- referentially transparent
end

For some purposes the precondition is disturbing. Therefore we de�ne
another query which does not have the precondition.

pred0: MATH_NAT
ensure

Result ˜ if is_zero then Current else pred end
end

The repeated application of the function pred will �nally retrieve the
initial object. Due to its precondition it cannot be applied to the initial
object. Therefore a chain of repeated applications of pred must terminate.
This makes it possible that pred can be used in recursion. This is valid
as long as all functions which produce new objects of type MATH_NAT never
create any cycles.

Repeated application of pred0 will not terminate. It �nally also encoun-
ters the initial object. But it can be in�nitely called on the initial object.

The predecessor function pred has two postconditions which need to be
veri�ed. The �rst one is used to derive the implementation and is therefore
valid implicitely. The second one states the pred is referential transparent.
This is implied by the �rst postcondition.

4 Recursion and proofs by induction

In this section we de�ne the recursive function minus and use it to de�ne
the order and equality relation for naturals. Furthermore we prove some
important properties.

Naturals numbers cannot be negative. Therefore our function minus has
to stop at zero.

minus alias "-" (other: MATH_NAT): MATH_NAT
ensure

Result ˜ if other.is_zero then
Current

else
minus(other.pred).pred0

end
end

The function minus (i.e. a � b) has a primitive recursive de�nition. If b
is zero then it returns a, because there is nothing to subtract. In case b is
not zero, it calculates the di�erence between a and the predecessor of b and
subtracts one from the result. For the subtraction of one the function pred0

is used, which subtracts only one, if the result is not yet zero.
I.e. in case that a is less or equal b the result is zero. In case that a is

greater than b then a positive value is returned. Due to this property the
function minus can be used to de�ned the order relation and equality.

is_less_equal alias "<=" (other: MATH_NAT): BOOLEAN
ensure

Result = (Current-other).is_zero
end

is_equal(other: MATH_NAT): BOOLEAN
ensure

Result = (Current<=other and other<=Current)
end

is_less alias "<" (other: MATH_NAT): BOOLEAN
ensure

Result = not (other<=Current)
end

is_greater alias ">" (other: MATH_NAT): BOOLEAN
ensure

Result = not (Current<=other)
end

is_greater_equal alias ">=" (other: MATH_NAT): BOOLEAN
ensure

Result = (other<=Current)
end

I have stated some properties of these functions based on the intuitive

understanding of natural numbers. But these properties must be proved for-
mally. We need to prove that is_equal de�nes an equivalence relation, i.e. it
is reexive, symmetric and transitive. We need to prove that is_less_equal
de�nes an order relation, i.e. it is reexive, antisymmetric and transitive.

The proof of these properties requires some intermediate steps. In the
following we state the properties to be proved as invariant properties (i.e. we
include them in then invariant of the class).

Proof of a � 0 = a As the �rst simple property we want to prove that
a� 0 = a. In terms our class this is the assertion

invariant
P1: all(a,b): b.is_zero => (a-b) ˜ a

The proof of this property is trivial since it just requires to expand the
de�nition of minus. Recall that the if expression

Result ˜ if other.is_zero then
Current

else
minus(other.pred).pred0

end

in minus is equivalent to

other.is_zero => Result ˜ Current
not other.is_zero => Result ˜ minus(other.pred).pred0

The �rst one is exactly the stated property.

Proof of (a-b).pred0˜(a.pred0-b) The reverse 0 � b = 0 is not that
trivial to prove, because it involves the recursive branch of the de�nition of
minus in case that b is not zero. We need the intermediate property

P2: all(a,b): (a-b).pred0 ˜ (a.pred0-b)

You might ask why this property is needed. If you try to prove 0 � b = 0
by induction, you will encounter expressions of the form (a-b).pred0 which
you cannot exploit further unless you have the stated property.

We prove P2 by induction on b. The initial case b is zero is simple, because
you have

b.is_zero => (a.pred0-b) ˜ a.pred0
b.is_zero => (a-b).pred0 ˜ a.pred0

due to the already proved property P1. The induction step requires us to
prove

not b.is_zero
=>
((a-b.pred).pred0 ˜ (a.pred0-b.pred)) -- Induction premise
=>
(a-b).pred0 ˜ (a.pred0-b)

We prove the induction step by the following equivalences

(a-b).pred0
˜
(a-b.pred).pred0.pred0 -- definition of "-"
˜
(a.pred0-b.pred).pred0 -- induction premise
˜
a.pred0-b -- definition of "-"

Proof of 0� b = 0 Now we can prove 0� b = 0, i.e.

P3: all(a,b): a.is_zero => (a-b).is_zero

We do this by induction on the variable b. The initial step is trivial since we
have already a� 0 = a available. The induction step requires us to prove

not b.is_zero
=>
(a.is_zero => (a-b.pred).is_zero) -- induction premise
=>
a.is_zero
=>
(a-b).is_zero

In order to prove that we use the equivalences

a-b
˜
(a-b.pred).pred0 -- definition of "-"
˜
(a.pred0-b.pred) -- P2
˜
(a-b.pred) -- a.is_zero and definition of pred0

This proves that a-b and a-b.pred are equivalent. Therefore a.is_zero and
the induction premise imply the required property.

Proof of a� a = 0 We can prove the property

P4: all(a): (a-a).is_zero

by induction. The initial step is trivial. The induction step requires us to
prove

not a.is_zero
=>
(a.pred-a.pred).is_zero
=>
(a-a).is_zero

We can use the equivalences

a-a
˜
(a-a.pred).pred0 -- definition of "-"
˜
a.pred0 - a.pred -- P2
˜
a.pred - a.pred -- not a.is_zero and definition of pred0

This allows us to conclude a� a = 0 from the induction premise.

Order and equivalence relation We have up to now in our reasoning just
used identity. This is because equality and order relation have not yet been
de�ned. Based of the function minus we have de�ned some other functions
which intuitively induce an equivalence relation and an order relation. But
the corresponding properties are neither stated nor proved. As a �rst step
we state the needed properties.

O1: all(a): a<=a
O2: all(a,b): (a<=b and b<=a) = (a=b)
O3: all(a,b,c): a<=b and b<=c => a<=c

E1: all(a): a=a
E2: all(a,b): (a=b) = (b=a)
E3: all(a,b,c): a=b and b=c => a=c

The properties O1, O2, E1 and E2 are almost trivial to prove. The ex-
pansion of the function de�nition of \�" or \=" immediately proves the
property. The transitivity properties O3 and E3 are not that trivial. Before
proving the transitivity we need to prove an intermediate assertion.

Proof of (a-b).is_zero => (a.pred0-b.pred0).is_zero First we state
the needed lemma.

P5: all(a,b):
(a-b).is_zero => (a.pred0-b.pred0).is_zero

The proof idea is obtained by expanding the de�nition of b.pred0. The
de�nition of this function introduces a case distinction on whether b is zero
or not.

In the case that b is zero we get a-b˜a and a.pred0-b.pred0˜a.pred0.
Using this substitutions and expanding the de�nition of a.pred0 proves im-
mediately the �rst case.

In the case that b is not zero we get the equivalence b.pred0˜b.pred and
a.pred0-b.pred0 is therefore equivalent to a-b using P2 and the de�nition
of minus. Inserting all that completes the proof.

Proof of transitivity The expansion of the de�nition of \�" in O3 yields
the assertion

all(a,b,c):
(a-b).is_zero and (b-c).is_zero => (a-c).is_zero

which is equivalent to

all(c): all(a,b):
(a-b).is_zero and (b-c).is_zero => (a-c).is_zero

The function minus is involved. This indicates that induction on the argu-
ment of minus can be tried. We start the proof by induction on c because
the expression a-c is involved in the consequent of the implication.

The initial step of the induction requires us to prove

c.is_zero
=>
all(a,b): (a-b).is_zero and (b-c).is_zero => (a-c).is_zero

The premise c.is_zero yields the equivalences b-c˜b and a-c˜a, i.e. we
have to prove

c.is_zero
=>
all(a,b): (a-b).is_zero and b.is_zero => a.is_zero

Using the commutativity of and and currying we get

c.is_zero
=>
all(a,b): b.is_zero

=>
(a-b).is_zero => a.is_zero

Now we can work on the inner implication and use the fact that b.is_zero
implies the identity a-b˜a which yields

c.is_zero
=>
all(a,b): b.is_zero

=>
a.is_zero => a.is_zero

Now the consequent of the inner implication is a triviality and the initial step
is proved.

The induction step requires us to prove

not c.is_zero
=>
(all(a,b):

(a-b).is_zero and (b-c.pred).is_zero => (a-c.pred).is_zero)
=>
(all(a,b): (a-b).is_zero and (b-c).is_zero

=>
(a-c).is_zero)

Using the de�nition of minus and P2 in the consequent of the inner im-
plication we get

not c.is_zero
=>
(all(x,y):

(x-y).is_zero and (y-c.pred).is_zero => (x-c.pred).is_zero)
=>
(all(a,b): (a-b).is_zero and (b-c).is_zero

=>
(a.pred0-c.pred).is_zero)

We have renamed the bound variables of the induction premise to see better
some pattern matching. The consequent of the inner implication matches
the consequent of the induction hypothesis if we substitute x by a.pred0.
In order to apply the induction premise we need to �nd a good substitution
of y. The �rst choice of using b does not lead to anything usable. However
using b.pred0 leads to

not c.is_zero
=>
(all(x,y):

(x-y).is_zero and (y-c.pred).is_zero => (x-c.pred).is_zero)
=>
(all(a,b): (a-b).is_zero and (b-c).is_zero

=>
(a.pred0-b.pred0).is_zero and
(b.pred0-c.pred).is_zero

The expression b.pred0-c.pred is identical to b-c using the de�nition of
minus and P2. With that the second conjunct can be derived from the an-
tecedent of the inner implication. In order to prove the �rst conjunct we
need property P5 which states that (a-b).is_zero implies the �rst conjunct.
With this derivation the induction proof is completed.

The transitivity property E3 of the equalitiy relation is proved by exand-
ing the de�nition of equality.

5 Creating non zero naturals

In the previous section we have assumed that repeated application of the
predecessor function pred �nally terminates. All recursive function de�ni-
tions and induction proofs are based on this crucial property. Therfore all
functions which create naturals have to maintain this property.

Non zero naturals are created using the creation procedure from_link

3. This creation procedure is not publicly available but functions can use
it to create non zero naturals. The most basic function of this type is the
successor function.

succ: MATH_NAT
ensure {NONE}

Result = MATH_NAT(immutable LINK(rep))
-- not referentially transparent!!

ensure
not Result.is_zero
Result.pred ˜ Current

end

This function has three postconditions. From the �rst one the imple-
mentation is derived and it is therefore valid by de�niton. Note that the
expression on the right hand side of the �rst postcondition is not referen-

tially transparent. Therefore the identity relation ˜ cannot be used in the
assertion.

The second postcondition can be proved easily. The creation procedure
from_link guarantees that is_zero will return the same value as the expres-
sion lnk˜Void. Since a non void argument is feed into the creation expression,
the resulting object is guaranteed to be non zero.

The third postcondition can be proved by the facts (see class LINK 2).

(immutable LINK(rep)).nxt ˜ rep

and

MATH_NAT(immutable LINK(rep)).pred ˜ MATH_NAT(rep)

(see speci�cation of from_link3 and pred3).
Now we have to prove that succ does not create cycles. This is easy to do.

The function creates an new immutable object of type LINK which points to
the attribute rep of the current object. Since the created LINK object is new,
no other existing LINK object can refer to it. Therefore if no cycle existed
before, then no cycles can exist after the creation.

Proof of P5: a�1+1 = a The third postcondition of pred states the equiv-
alence a.succ.pred˜a. From our intiuition we know that a.pred.succ=a

must be valid as well (provided that a is not zero). We want to prove this
property and state it formally

P5: all(a): not a.is_zero => a.pred.succ = a

Note that a.pred.succ and a cannot be identical, because the successor
function is not referentially transparent (it creates a new object on the heap).
We can only proof equality. The de�nition of is_equal requires us to prove

all(a): not a.is_zero => (a.pred.succ-a).is_zero
all(a): not a.is_zero => (a-a.pred.succ).is_zero

To prove the �rst one we use the following equivalences

(a.pred.succ-a).is_zero
˜
(a.pred.succ.pred0-a.pred).is_zero -- defintion of "-" and P2
˜
(a.pred.succ.pred-a.pred).is_zero -- definition of "pred0"
˜
(a.pred-a.pred).is_zero -- postcondition of "succ"
-- generally true(see P4)

13



The second one of the above assertions is proved by the equivalences

(a-a.pred.succ).is_zero
˜
(a-a.pred.succ.pred).pred0.is_zero -- definition of "-" and P2
˜
(a-a.pred).pred0.is_zero -- postconditon of "succ"
˜
(a-a).is_zero -- definition of "-"
-- generally true (see P4)

Proof of P6: a = b ) a + 1 = b + 1 The equality of two naturals shall
imply the equality of their successors, i.e.

P6: all(a,b): a=b => a.succ=b.succ

To prove that we use the equivalences

(a.succ-b.succ).is_zero
˜
(a.succ.pred0-b.succ.pred).is_zero -- definition of "-" and P2
˜
(a-b).is_zero -- definition of "pred0" and properties of "succ"

The last expression is implied by a=b. This proves that

all(a,b) a=b => (a-b).is_zero

Because of the symmetry of is_equal (E24) we can conclude

all(a,b) a=b => (b-a).is_zero

Both assertions prove P65.

Addition of natural numbers In the following we introduce more func-
tions which create non zero naturals. But all these functions will create new
LINK objects only through succ. Since succ maintains the property of no
cycles, all other function maintain these property implicitely.

The �rst function we introduce is addition

plus alias "+" (other: MATH_NAT): MATH_NAT
ensure

Result = if other.is_zero then Current
else plus(other.pred).succ
end

end

In the following we want to prove the usual properties like a+0 = a; 0+a =
a; a+ b = b+ a : : :.

Proof of P7: a+ 0 = a

P7: all(a,b): b.is_zero => a + b = a

This proof is trivial. It just requires to expand the de�nition of plus and use
the �rst branch.

Proof of P8: 0 + b = b

P8: all(a,b): a.is_zero => a+b = b

This property needs an induction proof with the induction variable b. The
initial step of the induction proof requires us to verify

all(a,b): b.is_zero => a.is_zero => a+b=b

Note that equality does not allow us to substitute the left hand side for the
right hand side (this is possible only with identity). But if we want to prove
x = z it is su�cient to prove x = y ^ y = z due to the transitivity property
E34 of equality. We can use that to prove a+b=b by proving a+b=a and a=b.
The �rst one is an immediate consequence of b.is_zero and P75. The second
one is implied by identity a˜b which is a consequence of a and b being zero.

The induction step requires us to prove

all(a,b): not b.is_zero
=>
(a.is_zero => (a+b.pred) = b.pred)
=>
a.is_zero
=>
a+b = b

i.e. we have to prove a+b=b under the given three hypotheses. We prove that
using the equalities

a+b
=
(a+b.pred).succ -- definition of "+"
=
b.pred.succ -- induction premise, P6
=
b -- P5

Proof of P9: a+ b = ((a� 1) + b) + 1 Formally stated:

all(a,b): not a.is_zero => a+b = (a.pred+b).succ

Proof by induction on b. The initial step of the induction proof is done using
the equalities

(a.pred+b).succ
=
a.pred.succ -- P6 and P7
=
a -- P5
=
a+b -- P7

The induction step requires to prove

all(a,b): not b.is_zero
=>
(not a.is_zero => a+b.pred = (a.pred+b.pred).succ)
=>
not a.is_zero
=>
a+b = (a.pred+b).succ

In order to prove the conclusion we use the equalities

(a.pred+b).succ
=
(a.pred+b.pred).succ.succ -- definition of "+" and P6
=
(a+b.pred).succ -- induction premise and P6
=
a+b -- definition of "+"

Proof of commutativity P10: a+ b = b+ a With the above properties
we can prove the commutativity of addition which is formally the assertion

all(a,b): a+b = b+a

The proof uses induction on the variable b. The initial step of the induction
is an immediate consequence of P8 : a + 0 = a and P9 : 0 + a = a. The
induction step requires us to prove

all(a,b): not b.is_zero
=>
(a+b.pred) = (b.pred+a)
=>
a+b = b+a

This is done using the equalities

a+b
=
(a+b.pred).succ -- definition of "+"
=
(b.pred+a).succ -- induction premise and P6
=
b+a -- P9

6 Some more properties

In the previous sections all proofs have been given in detail. In this section
we give some additional properties and give just the proof outline.

Associativity (a+ b) + c = a+ (b+ c)

all(a,b,c): check
(a+b)+c = a+(b+c) -- associativity
a=b => a+c = b+c -- intermediate property 1
(a+b).succ = a+b.succ -- " " 2

end

The associativity is proved by induction on c using the intermediate proper-
ties 1 and 2. The intermediate property 1 is proved by induction on c and the
intermediate property is proved by induction on b. All proofs rely heavily on
the transitivity of equality. The intermediate property 1 is a generalization
of the already proved property P6 5 that equality of two values implies the
equality of their successors.

De�niton of product

product alias "*" (other: MATH_NAT): MATH_NAT
ensure

Result = if other.is_zero then other
else Current + Current*other.pred end

end

Commutativity of product

all(a,b,c): check
a1: a*b = b*a -- induction on b
a2: b.is_zero => (a*b).is_zero -- trivial
a3: a.is_zero => (a*b).is_zero -- induction on b

a4: a=b => a*c = b*c -- induction on c
a5: a*b + b = a.succ*b -- induction on b
a6: not b.is_zero

=> a.succ + b.pred = a+b -- use a7
a7: (a+b).succ = a.succ + b -- induction on b

end

Nearly all of these properties can be proved by induction using some of the
other properties. The programmer just has to state the properties, the proof
engine of Modern Ei�el does the actual proof (or complains, if it fails to
prove).

Distributivity of product over addition ab+ ac = a(b+ c)

all(a,b,c): a*b + a*c = a*(b+c) -- induction on c

Associativity of product (ab)c = a(bc)

all(a,b,c): (a*b)*c = a*(b*c) -- induction on c

De�nition of power ab

zero: MATH_NAT = MATH_NAT
one: MATH_NAT = zero.succ

power alias "ˆ"(other: MATH_NAT): MATH_NAT
ensure

Result = if other.is_zero then one
else Current*Currentˆ(other.pred) end

end

Property ab+c = abac

all(a,b,c): check
aˆ(b+c) = aˆb * aˆc -- induction on c

b=c => aˆb = aˆc -- induction on c
check

18


	Introduction
	The class LINK
	Outline of the class MATH_NAT
	Recursion and proofs by induction
	Creating non zero naturals
	Some more properties

